Makerguides.com

How to control a stepper motor with DRV8825 driver and Arduino

Written by Benne de Bakker (https://www.makerguides.com/author/benne-de-bakker/)



This article includes everything you need to know about controlling a stepper motor with the DRV8825 stepper motor driver and Arduino. I have included a wiring diagram, a tutorial on how to set the current limit and many example codes.

Although you can use this driver without an Arduino library, I highly recommend you also take a look at the example code for the **AccelStepper library** at the end of this tutorial. This library is fairly easy to use and can greatly improve the performance of your hardware.

After each example, I break down and explain how the code works, so you should have no problems modifying it to suit your needs.

If you would like to learn more about other stepper motor drivers, then the articles below might be useful:

- How to control a stepper motor with A4988 driver and Arduino (https://www.makerguides.com/a4988-stepper-motor-driver-arduino-tutorial/)
- 28BYJ-48 Stepper Motor with ULN2003 Driver and Arduino Tutorial (https://www.makerguides.com/28byj-48-stepper-motor-arduino-tutorial/)
- How to control a Stepper Motor with Arduino Motor Shield Rev3 (https://www.makerguides.com/arduino-motor-shield-stepper-motor-tutorial/)

About the driver

At the heart of the driver you will find a chip made by Texas Instruments: the DRV8825 Stepper Motor Controller IC. This integrated motor driver makes interfacing with a microcontroller super easy as you only need two pins to control both the speed and the direction of the stepper motor.

The driver has a maximum output capacity of 45V and ± 2 A which is great for driving small to medium sized stepper motors like a NEMA 17 (https://amzn.to/2IMe92B) bipolar stepper motor.

The chip has several safety functions built-in like overcurrent, short circuit, under voltage lockout and over temperature protection. You can find more specifications in the table below.

DRV8825 Specifications

Minimum operating voltage	8.2 V
Maximum operating voltage	45 V
Continuous current per phase	1.5 A
Maximum current per phase	2.2 A
Minimum logic voltage	2.5 V
Maximum logic voltage	5.25 V
Microstep resolution	full, 1/2, 1/4, 1/8, 1/16 and 1/32
Reverse voltage protection?	No
Dimensions	15.5 × 20.5 mm (0.6″ × 0.8″)

For more information you can check out the datasheet here.

DRV8825 Datasheet **(https://www.makerguides.com/wp-content/uploads/2019/02/DRV8825-Datasheet.pdf)**

Differences between DRV8825 and A4988

The DRV8825 is quite similar to the A4988 (https://amzn.to/2BKd2vq) but there are some key differences:

- The DRV8825 offers 1/32 microstepping, whereas the A4988 only goes down to 1/16-step. Higher microstepping results in smoother, quieter operation but is not always needed.
- The current limit potentiometer is at a different location
- The relation between the reference voltage and the current limit is different.
- The DRV8825 requires a minimum STEP pulse duration of 1.9µs, the A4988 requires 1µs minimum.

- The DRV8825 can be used with higher voltage motor power supply (45 V vs 35 V). This means it is less susceptible to damage from LC voltage spikes.
- The DRV8825 can deliver slightly more current than the A4988 without any additional cooling.

Note that the pinout of the DRV8825 is exactly the same as for the A4988, so it can be used as a drop in replacement!



A4899 (left) vs DRV8825 (right)

Microstep settings

Stepper motors typically have a step size of 1.8° or 200 steps per revolution, this refers to full steps. A microstepping driver such as the DRV8825 allows higher resolutions by allowing intermediate step locations. This is achieved by energizing the coils with intermediate current levels.

For instance, driving a motor in quarter-step mode will give the 200-step-per-revolution motor 800 microsteps per revolution by using four different current levels.



DRV8825 Pinout

The resolution (step size) selector pins (M0, M1, and M2) allow you to select one of the six step resolutions according to the table below.

DRV8825 microstep settings

M0	M1	M2	Microstep resolution
Low	Low	Low	Full step
High	Low	Low	1/2 step
Low	High	Low	1/4 step
High	High	Low	1/8 step
Low	Low	High	1/16 step
High	Low	High	1/32 step
Low	High	High	1/32 step
High	High	High	1/32 step

All three inputs have internal 100kΩ pull-down resistors, so leaving the three microstep selection pins disconnected results in full-step mode.

I often use a CNC-shield (https://amzn.to/2Ebtdnb) or expansion board in combination with these drivers. The expansion board has 3 dip switches to set MS1 – MS3 high or low and on the CNC-shield you can install jumpers. If you are using the driver with a breadboard, you can just use jumper wires to connect the selector pins to 5V.

Things used in this tutorial:

If you want to follow this tutorial, you will need the following components:

Hardware components

Arduino UNO R3 (https://www.amazon.com/Arduino-Starter-Kit-English-Official/dp/B009UKZV0A/ref=as_li_ss_tl?keywords=arduino+uno&qid=1561972887&s=gateway&sr=8-10&linkCode=ll1&tag=makerguides-20&linkId=98f3d35e6867ccc54d7dd66e23dfc216&language=en_US) If you want to control many stepper motors, you can use an Arduino Mega (https://amzn.to/2J3gbuz).	x 1	Amazon (https://www.amazon.com/Arduino-Starter-Kit-Englis Official/dp/B009UKZV0A/ref=as_li_ss_t keywords=arduino+uno&qid=1561972887&s=gateway&sr= 10&linkCode=ll1&tag=makerguide 20&linkId=98f3d35e6867ccc54d7dd66e23dfc216&language=en_U
DRV8825 stepper motor driver (https://amzn.to/2TO0gmz)	x 1	Amazon (https://amzn.to/2TO0gm
NEMA 17 stepper motor (https://amzn.to/2V0gjhk)	x 1	Amazon (https://amzn.to/2V0gjh
Breadboard (https://amzn.to/2MQHicc) I highly recommend to buy at least 1 good quality breadboard like the BusBoard Prototype Systems BB400 (https://amzn.to/2MQHicc) or BB830 (https://amzn.to/2GM261Q).	x 1	Amazon (https://amzn.to/2MQHic
Jumper wires (https://amzn.to/2DCt2R8)	~ 10	Amazon (https://amzn.to/2DCt2R
USB Type-B cable (https://amzn.to/2GR73aX)	x 1	Amazon (https://amzn.to/2GR73a
100µF capacitor (https://amzn.to/2tuuuj3)	x 1	Amazon (https://amzn.to/2tuuuj
Power supply (8-35V) (https://amzn.to/2WZBhOZ)	x 1	Amazon (https://amzn.to/2WZBhO

Tools

Multimeter (https://amzn.to/2EXUidl)	Amazon (https://amzn.to/2EXUidl)
Small screwdriver (https://amzn.to/2X2ZtQH) I like the Wiha PicoFinish (https://amzn.to/2X2ZtQH).	Amazon (https://amzn.to/2X2ZtQH)
Alligator test leads (https://amzn.to/2E9uj2F) (Optional but very handy)	Amazon (https://amzn.to/2E9uj2F)

Software

Arduino IDE (https://www.arduino.cc/en/Main/Software)

As I mentioned earlier, I like to use this driver in combination with a CNC-shield (https://amzn.to/2Ebtdnb) or expansion board (https://amzn.to/2SOi1EZ). Such a shield already includes capacitors and offers an easy way to select the microstepping resolution. It makes wiring much easier and is a great option if you need a more permanent solution than a breadboard.

Wiring – Connecting DRV8825 to Arduino and stepper motor



Wiring diagram/schematic for DRV8825 stepper motor driver with Arduino and stepper motor.

The wiring diagram/schematic above shows you how to connect the DRV8825 driver to a stepper motor and the Arduino.

The connections are also given in the following table:

DRV8825 Connections

DRV8825	Connection
VMOT	8.2-45V
GND	Motor ground
SLP	5V
RST	5V
GND	Logic ground
STP	Pin 3
DIR	Pin 2
A1, A2, B1, B2	Stepper motor

- The motor power supply is connected to GND and VMOT (top right).
- The two coils of the stepper motor are connected to A1, A2 and B1, B2 (see below).
- The GND pin (lower right) is connected to the ground pin of the microcontroller and VDD is connected to 5V.
- The STP (step) and DIR (direction) pin are connected to digital pin 3 and 2 respectively. You can choose a different digital pin if you want, but these are the ones I used for this tutorial and the example code.
- You need to connect RST (reset) and SLP (sleep) to 5V, otherwise the driver won't turn on.
- The EN (enable) pin can be left disconnected, it is pulled low by default. When this pin is set high the driver is disabled.
- The DRV8825 also features a FAULT output that drives low whenever the H-bridge FETs are disabled as the result of over-current protection or thermal shutdown. This pin is left disconnected for this tutorial.

In the rest of this tutorial **I have left MS1, MS2 and MS3 disconnected**, so the driver operates in **full step mode**. This makes explaining the code a bit easier. Normally I would use 1/16 or 1/32 microstepping and connect the appropriate pins to 5V (see table in introduction).

Warning

The DRV8825 carrier board uses low-ESR ceramic capacitor, which makes it susceptible to destructive LC voltage spikes, especially when using power leads longer than a few inches.

To protect the driver you can connect an electrolytic capacitor between VMOT and GND. Pololu suggests a capacitor of 47µF or more (I used a 100 µF capacitor). I like these assortment boxes (https://amzn.to/2LXWTKk) from Amazon, this way I always have some capacitors of the right size on hand.

How to determine the correct stepper motor wiring?

If you can't find the datasheet of your stepper motor, it can be difficult to figure out how to wire your motor correctly. I use the following trick to determine how to connect 4 wire bipolar stepper motors:

The only thing you need to identify is the **two pairs of wires** which are connected to the two coils of the motor. The wires from one coil get connected to 1A and 1B and the other to 2A and 2B, the polarity doesn't matter.

To find the two wires from one coil, do the following with the motor disconnected:

1. Try to spin the shaft of the stepper motor by hand and notice how hard it is to turn.
2. Now pick a random pair of wires from the motor and touch the bare ends together.
3. Next, try to spin the shaft of the stepper motor again.

If you feel a lot of resistance, you have found a pair of wires from the same coil. If you can spin the shaft freely, try another pair of wires. Now connect the two coils to the pins shown in the wiring diagram above.

(If it is still unclear, please leave a comment below, more info can also be found on the RepRap.org wiki (https://reprap.org/wiki/Stepper_wiring))

How to set the current limit?

Before you start programming your Arduino and start using the driver there is one very **important** thing you need to do that **a lot of people forget**: set the current limit!

This step is not very complicated but absolutely necessary to protect your stepper motor and the driver. If you do not set an appropriate current limit, your motor can draw more current than it or your driver can handle, this is likely to damage one or both of them.

To set the current limit you need to measure a reference voltage and adjust the on-board potentiometer accordingly. You will need a small screwdriver, a multimeter to measure the reference voltage and alligator test leads (optional but very handy).



Current limit wiring diagram for DRV8825 driver.

To measure the reference voltage, the driver needs to be powered. The DRV8825 only needs power via VMOT (8.2-45V) and you need to apply 5V to RST and SLP, otherwise the driver won't turn on. It's best to disconnect the stepper motor while you do this.

If you have already wired up the driver as I have shown before, you can leave the Arduino connected to power the RST and SLP pins.

Required connections to set the current limit

DRV8825	Connection
VMOT	8.2-45 V
GND	Motor ground
SLP	5V
RST	5V
GND	Logic ground

Current limit formula

The next step is to calculate the current limit with the following formula:

$$\text{Current Limit} = \text{Vref} \times 2$$

So this means that for a current limit of 1A the Vref should be 0.5V.

To select the right current limit, take a look at the datasheet of your stepper motor. If you can't find the current rating of your motor, I recommend starting with a current limit of 1A. You can always increase it later if your motor/driver is missing steps.

Bonus info: When using the driver in full-step mode, the current through each coil is limited to approximately 70% of the set current limit. This means that you would need to set the current limit 40% higher or 1.4 A in full-step mode. When using microstepping, the formula above applies.

If your motor is making a lot of noise, try to lower the current limit. It's best to set the current limit just high enough so the motor doesn't miss steps.

Measuring Vref

Now you will need to measure the reference voltage (Vref) between the two points marked on the picture below (GND and the potentiometer) and adjust it to the value you calculated.



Vref probe points (GND and potentiometer).

I recommend using alligator test leads (https://amzn.to/2E9uj2F) clamped to the screwdriver to set the current limit. This allows you to adjust the potentiometer and measure the reference voltage at the same time.

Note: There is another way to measure the current limit and that is to directly measure the current draw of the stepper motor. Personally I find the above method a lot easier.

Pololu (https://www.pololu.com/)mentions the following on their website:

"Note: The coil current can be very different from the power supply current, so you should not use the current measured at the power supply to set the current limit. The appropriate place to put your current meter is in series with one of your stepper motor coils."

Current limit FAQ

Do I need to have the stepper motor connected or not?
No, you don't need to connect te stepper motor to the driver when setting the current limit. To be on the safe side, disconnect your motor, it sometimes interferes with measuring the Vref voltage.

Do I need to turn the motor by running the Arduino motor sketch?
No, see question above.

Do I need to turn the potentiometer clock- or counterclockwise to raise Vref?
This depends on the manufacturer of the driver. If you have genuine Polulu breakout boards of the DRV8825 or A4988 you turn the potentiometer clockwise to raise Vref and counterclockwise to lower it.

Cooling the driver

The DRV8825 driver IC has a maximum current rating of 2.5 A per coil, but without a heat sink it can only supply about 1.5 A per coil before it starts to overheat.

The driver usually comes with a small adhesive-backed heat sink, which I recommend you to install right away. You can also buy a bunch of small heat sinks (https://amzn.to/2VPaLLx) from Amazon for really cheap.

Basic Arduino example code to control a stepper motor

Now that you have wired up the driver and set the current limit, it is time to connect the Arduino to the computer and upload some code. You can upload the following example code to your Arduino using the Arduino IDE (https://www.arduino.cc/en/main/software). For this specific example you do not need to install any libraries.

This sketch controls both the speed, the number of revolutions and the spinning direction of the stepper motor.

You can open the code in a new window by clicking on the button in the top right corner.

```
/*Example sketch to control a stepper motor with A4988/DRV8825 stepper motor driver and Arduino without a library. More info: https://www.makerguides.com */

// Define stepper motor connections and steps per revolution:
#define dirPin 2
#define stepPin 3
#define stepsPerRevolution 200

void setup() {
  // Declare pins as output:
  pinMode(stepPin, OUTPUT);
  pinMode(dirPin, OUTPUT);
}

void loop() {
  // Set the spinning direction clockwise:
  digitalWrite(dirPin, HIGH);

  // Spin the stepper motor 1 revolution slowly:
  for (int i = 0; i < stepsPerRevolution; i++) {
    // These four lines result in 1 step:
    digitalWrite(stepPin, HIGH);
    delayMicroseconds(2000);
    digitalWrite(stepPin, LOW);
    delayMicroseconds(2000);
  }
```

```
26.
27.    delay(1000);
28.
29.    // Set the spinning direction counterclockwise:
30.    digitalWrite(dirPin, LOW);
31.
32.    // Spin the stepper motor 1 revolution quickly:
33.    for (int i = 0; i < stepsPerRevolution; i++) {
34.      // These four lines result in 1 step:
35.      digitalWrite(stepPin, HIGH);
36.      delayMicroseconds(1000);
37.      digitalWrite(stepPin, LOW);
38.      delayMicroseconds(1000);
39.    }
40.
41.    delay(1000);
42.
43.    // Set the spinning direction clockwise:
44.    digitalWrite(dirPin, HIGH);
45.
46.    // Spin the stepper motor 5 revolutions fast:
47.    for (int i = 0; i < 5 * stepsPerRevolution; i++) {
48.      // These four lines result in 1 step:
49.      digitalWrite(stepPin, HIGH);
50.      delayMicroseconds(500);
51.      digitalWrite(stepPin, LOW);
52.      delayMicroseconds(500);
53.    }
54.
55.    delay(1000);
56.
57.    // Set the spinning direction counterclockwise:
58.    digitalWrite(dirPin, LOW);
59.
60.    //Spin the stepper motor 5 revolutions fast:
61.    for (int i = 0; i < 5 * stepsPerRevolution; i++) {
62.      // These four lines result in 1 step:
63.      digitalWrite(stepPin, HIGH);
64.      delayMicroseconds(500);
65.      digitalWrite(stepPin, LOW);
66.      delayMicroseconds(500);
67.    }
68.
69.    delay(1000);
70.  }
```

How the code works:

The sketch starts with defining the step and direction pins. I connected them to Arduino pin 3 and 2.

The statement `#define` is used to give a name to a constant value. The compiler will replace any references to this constant with the defined value when the the program is compiled. So everywhere you mention `dirPin`, the compiler will replace it with the value 2 when the program is compiled.

I also defined a `stepsPerRevolution` constant. Because I set the driver to full step mode I set it to 200 steps per revolution. Change this value if your setup is different.

```
3.  // Define stepper motor connections and steps per revolution:
4.  #define dirPin 2
5.  #define stepPin 3
6.  #define stepsPerRevolution 200
```

In the `setup()` section of the code, all the motor control pins are declared as digital OUTPUT with the function `pinMode()`.

```
8.  void setup() {
9.    // Declare pins as output:
```

```
  pinMode(stepPin, OUTPUT);
  pinMode(dirPin, OUTPUT);
}
```

In the `loop()` section of the code, we let the motor spin one revolution slowly in the CW direction and one revolution quickly in the CCW direction. Next we let the motor spin 5 revolutions in each directions with a high speed. So how do you control the speed, spinning direction and number of revolutions?

```
  // Set the spinning direction clockwise:
  digitalWrite(dirPin, HIGH);

  // Spin the stepper motor 1 revolution slowly:
  for(int i = 0; i < stepsPerRevolution; i++)
  {
    // These four lines result in 1 step:
    digitalWrite(stepPin, HIGH);
    delayMicroseconds(2000);
    digitalWrite(stepPin, LOW);
    delayMicroseconds(2000);
  }
```

Control spinning direction:

To control the spinning direction of the stepper motor we set the DIR (direction) pin either HIGH or LOW. For this we use the function `digitalWrite()` . Depending on how you connected the stepper motor, setting the DIR pin high will let the motor turn CW or CCW.

Control number of steps or revolutions:

In this example sketch, the for loops (https://www.arduino.cc/reference/en/language/structure/control-structure/for/) control the number of steps the stepper motor will take. The code within the for loop results in 1 step of the stepper motor. Because the code in the loop is executed 200 times (stepsPerRevolution), this results in 1 revolution. In the last two loops, the code within the for loop is executed 1000 times, which results in 1000 steps or 5 revolutions.

Note that you can change the second term in the for loop to whatever number of steps you want. `for(int i = 0; i < 100; i++)` would result in 100 steps, or half a revolution.

Control speed:

The speed of the stepper motor is determined by the frequency of the pulses we send to the STEP pin. The higher the frequency, the faster the motor runs. You can control the frequency of the pulses by changing `delayMicroseconds()` in the code. The shorter the delay, the higher the frequency, the faster the motor runs.

AccelStepper library tutorial

The AccelStepper library written by Mike McCauley is an awesome library to use for your project. One of the advantages is that it supports acceleration and deceleration, but it has a lot of other nice functions too.

You can download the latest version of this library here (https://www.airspayce.com/mikem/arduino/AccelStepper/index.html) or click the button below.

AccelStepper-1.59.zip **(https://www.makerguides.com/wp-content/uploads/2019/02/AccelStepper-1.59.zip)**

You can install the library by going to **Sketch > Include Library > Add .ZIP Library...** in the Arduino IDE.

Another option is to navigate to **Tools > Manage Libraries...** or type Ctrl + Shift + I on Windows. The Library Manager will open and update the list of installed libraries.



You can search for **'accelstepper'** and look for the library by Mike McCauley. Select the latest version and then click Install.



1. Continuous rotation example code

The following sketch can be used to run one or more stepper motors continuously at a constant speed. (No acceleration or deceleration is used).

```
/*Example sketch to control a stepper motor with DRV8825 stepper motor driver, AccelStepper library and Arduino: continuous rotation. More info: https://www.makerguides.com */

// Include the AccelStepper library:
#include <AccelStepper.h>

// Define stepper motor connections and motor interface type. Motor interface type must be set to 1 when using a driver:
#define dirPin 2
#define stepPin 3
#define motorInterfaceType 1

// Create a new instance of the AccelStepper class:
AccelStepper stepper = AccelStepper(motorInterfaceType, stepPin, dirPin);

void setup() {
  // Set the maximum speed in steps per second:
  stepper.setMaxSpeed(1000);
}

void loop() {
  // Set the speed in steps per second:
  stepper.setSpeed(400);
  // Step the motor with a constant speed as set by setSpeed():
  stepper.runSpeed();
}
```

How the code works:

The first step is to include the library with `#include <AccelStepper.h>` .

```
// Include the AccelStepper library:
#include <AccelStepper.h>
```

The next step is to define the DRV8825 to Arduino connections and the motor interface type. The motorinterface type must be set to 1 when using a step and direction driver. You can find the other interface types here (https://www.airspayce.com/mikem/arduino/AccelStepper/classAccelStepper.html#a608b2395b64ac15451d16d0371fe13ce).

The statement `#define` is used to give a name to a constant value. The compiler will replace any references to this constant with the defined value when the the program is compiled. So everywhere you mention `dirPin` , the compiler will replace it with the value 2 when the program is compiled.

```
// Define stepper motor connections and motor interface type. Motor interface type must be set to 1 when using a driver:
#define dirPin 2
#define stepPin 3
#define motorInterfaceType 1
```

Next, you need to create a new instance of the AccelStepper class with the appropriate motor interface type and connections.

In this case I called the stepper motor ‘stepper’ but you can use other names as well, like ‘z_motor’ or ‘liftmotor’ etc. `AccelStepper liftmotor = AccelStepper(motorInterfaceType, stepPin, dirPin);` . The name that you give to the stepper motor will be used later to set the speed, position and acceleration for that particular motor. You can create multiple instances of the AccelStepper class with different names and pins. This allows you to easily control 2 or more stepper motors at the same time.

```
// Create a new instance of the AccelStepper class:
AccelStepper stepper = AccelStepper(motorInterfaceType, stepPin, dirPin);
```

In the `setup()` section of the code we define the maximum speed in steps/second. Speeds of more than 1000 steps per second can be unreliable, so I set this as the maximum. Note that I specify the name of the stepper motor ('stepper'), for which I want to define the maximum speed. If you have multiple stepper motors connected, you can specify a different speed for each motor:

```
14. void setup() {
15.   // Set the maximum speed in steps per second:
16.   stepper.setMaxSpeed(1000);
17.   stepper2.setMaxSpeed(500);
18. }
```

In the `loop()` we first set the speed that we want the motor to run at. For this we use the function `setSpeed()`. (you can also place this in the setup section of the code).

`stepper.runSpeed()` polls the motor and when a step is due, executes 1 step. This depends on the set speed and the time since the last step. If you want to change the direction of the motor, you can set a negative speed: `stepper.setSpeed(-400);` turns the motor the other way.

```
19. void loop() {
20.   // Set the speed in steps per second:
21.   stepper.setSpeed(400);
22.   // Step the motor with a constant speed as set by setSpeed():
23.   stepper.runSpeed();
24. }
```


2. Example code to control number of steps or revolutions

To let the motor rotate a specific number of steps I prefer to use a while loop (https://www.arduino.cc/reference/en/language/structure/control-structure/while/) in combination with `stepper.currentPosition()`. You can use the following example code, to let the motor run back and forth.

```
1.  /*Example sketch to control a stepper motor with DRV8825 stepper motor driver, AccelStepper library and Arduino:
    number of steps or revolutions. More info: https://www.makerguides.com */
2.
3.  // Include the AccelStepper library:
4.  #include <AccelStepper.h>
5.
6.  // Define stepper motor connections and motor interface type. Motor interface type must be set to 1 when using a
    driver:
7.  #define dirPin 2
8.  #define stepPin 3
9.  #define motorInterfaceType 1
10.
11. // Create a new instance of the AccelStepper class:
12. AccelStepper stepper = AccelStepper(motorInterfaceType, stepPin, dirPin);
13.
14. void setup() {
15.   // Set the maximum speed in steps per second:
16.   stepper.setMaxSpeed(1000);
17. }
18.
19. void loop() {
```

```
20.    // Set the current position to 0:
21.    stepper.setCurrentPosition(0);
22.
23.    // Run the motor forward at 200 steps/second until the motor reaches 400 steps (2 revolutions):
24.    while(stepper.currentPosition() != 400)
25.    {
26.      stepper.setSpeed(200);
27.      stepper.runSpeed();
28.    }
29.
30.    delay(1000);
31.
32.    // Reset the position to 0:
33.    stepper.setCurrentPosition(0);
34.
35.    // Run the motor backwards at 600 steps/second until the motor reaches -200 steps (1 revolution):
36.    while(stepper.currentPosition() != -200)
37.    {
38.      stepper.setSpeed(-600);
39.      stepper.runSpeed();
40.    }
41.
42.    delay(1000);
43.
44.    // Reset the position to 0:
45.    stepper.setCurrentPosition(0);
46.
47.    // Run the motor forward at 400 steps/second until the motor reaches 600 steps (3 revolutions):
48.    while(stepper.currentPosition() != 600)
49.    {
50.      stepper.setSpeed(400);
51.      stepper.runSpeed();
52.    }
53.
54.    delay(3000);
55.  }
```

Code explanation:

The first part of the code up to the loop() section is exactly the same as in the previous example.

In the loop I make use of a while loop in combination with the `currentPosition()` function. First, I set the current position of the stepper motor to zero with `stepper.setCurrentPosition(0)` .

```
20.    // Set the current position to 0:
21.    stepper.setCurrentPosition(0);
```

Next we make use of the while loop. A while loop will loop continuously, and infinitely, until the expression inside the parenthesis, () becomes false. So in this case I check if the current position of the stepper motor is not equal to 400 steps (!= means: is not equal to). While this is not the case, we run the stepper motor at a constant speed as set by `setSpeed()` .

```
22.    // Run the motor forward at 200 steps/second until the motor reaches 400 steps (2 revolutions):
23.    while(stepper.currentPosition() != 400)
24.    {
25.      stepper.setSpeed(200);
26.      stepper.runSpeed();
27.    }
```

In the rest of the loop, we do exactly the same, just with a different speed and target position.

3. Acceleration and deceleration example code

With the following sketch you can add acceleration and deceleration to the movements of the stepper motor, without any complicated coding. In the following example, the motor will run back and forth with a speed of 200 steps per second and an acceleration of 30 steps per second per second.

```
/*Example sketch to control a stepper motor with DRV8825 stepper motor driver, AccelStepper library and Arduino: acceleration and deceleration. More info: https://www.makerguides.com */

// Include the AccelStepper library:
#include <AccelStepper.h>

// Define stepper motor connections and motor interface type. Motor interface type must be set to 1 when using a driver:
#define dirPin 2
#define stepPin 3
#define motorInterfaceType 1

// Create a new instance of the AccelStepper class:
AccelStepper stepper = AccelStepper(motorInterfaceType, stepPin, dirPin);

void setup() {
  // Set the maximum speed and acceleration:
  stepper.setMaxSpeed(200);
  stepper.setAcceleration(30);
}

void loop() {
  // Set the target position:
  stepper.moveTo(600);
  // Run to target position with set speed and acceleration/deceleration:
  stepper.runToPosition();

  delay(1000);

  // Move back to zero:
  stepper.moveTo(0);
  stepper.runToPosition();

  delay(1000);
}
```

Code explanation:

In the setup(), besides the maximum speed, we need to define the acceleration/deceleration. For this we use the function `setAcceleration()` .

```
void setup() {
```

```
  // Set the maximum speed and acceleration:
  stepper.setMaxSpeed(200);
  stepper.setAcceleration(30);
}
```

In the loop section of the code, I used a different way to let the motor rotate a predefined number of steps. The function `stepper.moveTo()` is used to set the target position. The function `stepper.runToPostion()` moves the motor (with acceleration/deceleration) to the target position and blocks until it is at the target position. Because this function is blocking, you shouldn't use this when you need to control other things at the same time.

```
  // Set the target position:
  stepper.moveTo(600);
  // Run to target position with set speed and acceleration/deceleration:
  stepper.runToPosition();
```

Conclusion

In this article I have shown you how to control a stepper motor with the DRV8825 stepper motor driver and Arduino. I hope you found it useful and informative. If you did, please **share it with a friend** who also likes electronics and making things!

I have personally used this driver a lot for a bunch of 3D printers and other CNC related projects but I would love to know what projects you plan on building (or have already built) with this driver. If you have any questions, suggestions or if you think that things are missing in this tutorial, **please leave a comment down below**.

Note that comments are held for moderation in order to prevent spam.

Beginner

Comments

Olivier says
May 9, 2019 at 5:49 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-35)

Hi,
I'm currently researching for an automated blinds project using this driver and some stepper motors. I want to use an RTC to have the blinds open and close at certain times and also have an IR remote to manually operate them. I'm currently simulating it using a simple circuit with a pushbutton but I'm having a hard time with the code. I basically press the button, I want it to move a certain number of rotations (predetermined to the closed position.) and if I press again, move the other way back to the open position. Do you have any input on how I should set the code up?
Thank you, this is an amazing ressource.
Olivier

Reply

> Benne de Bakker says
> May 9, 2019 at 8:06 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-38)
>
> Hi Olivier,
>
> Thank you for your comment. I wrote a simple addition to one of the example sketches above. You could use the following sketch in combination with a momentary push button. The push button is connected between GND and digital pin 4. When you press the button, the motor moves from the 0 position to the open position (400). And when you press it again it moves back to the 0 position. Hopefully this gives you some inspiration for the rest of your code. Good luck with the rest of your project!
>
> Benne
>
> Code:
>
> https://www.makerguides.com/wp-content/uploads/2019/05/DRV8825-example-sketch-with-AccelStepper-library-and-push-button.pdf (https://www.makerguides.com/wp-content/uploads/2019/05/DRV8825-example-sketch-with-AccelStepper-library-and-push-button.pdf)
>
> Reply

bryan (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/) says
May 14, 2019 at 9:04 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-48)

is ist posible to add another stepper motor ? how to do it?

Reply

> Benne de Bakker says
> May 14, 2019 at 9:13 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-49)
>
> Hi,

The DRV8825 can only drive one stepper motor, but you can add multiple DRV8825 drivers to the Arduino. You will need to pick a different step and direction pin for the second driver.

In the code you can create a second stepper motor object with a different name, like stepper2. AccelStepper stepper2(1,stepPin2,dirPin2); See also the code explanation under the first AccelStepper example. In the rest of the code you can control the second stepper motor by using the new name in the rest of the functions: stepper2.setSpeed(200); and stepper2.runSpeed(); for example.

In this way you can control both motors separately.

Benne

Reply

bryan (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/?unapproved=48&moderation-hash=8786faa662b7672586fd3be805cb5b7e#comment-48) says
May 14, 2019 at 11:52 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-51)

is it okay to use A4998 driver instead of DRV8825? is it the same schematic driagram and add just add driver and stepper motor ? thanks for the reply .

Reply

Benne de Bakker says
May 14, 2019 at 12:01 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-52)

Yes, the A4988 has the exact same pinout, so it can be used as a drop in replacement for the DRV8825. See my tutorial about the A4988 here (code is the same): https://www.makerguides.com/a4988-stepper-motor-driver-arduino-tutorial/ (https://www.makerguides.com/a4988-stepper-motor-driver-arduino-tutorial/)

Reply

bryan (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/?unapproved=51&moderation-hash=f2e4279e04ef588fc64dfca8d0e92155#comment-51) says
May 14, 2019 at 12:06 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-53)

thanks! it helps me a lot!

bryan (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/?unapproved=51&moderation-hash=f2e4279e04ef588fc64dfca8d0e92155#comment-51) says
May 14, 2019 at 12:08 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-54)

if im going to add stepper motor and driver should i also add capacitor?

Benne de Bakker says
May 15, 2019 at 3:39 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-60)

> It's best to add a capacitor between GND and VMOT for each driver. The CNC shield, like this one (https://amzn.to/2YAaGrZ), also has one capacitor per driver.

Mike Sims (http://www.facebook.com/michael.dean.sims) says
June 12, 2019 at 10:03 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-98)

Hello, thank you for the excellent and thorough writeup!

You say that the DRV8825 can microstep up to 1/32 but under the section entitled, "Microstep Settings", you only show how to set pins M0, M1, M2 up to 1/16 … and I'm thinking since 1/32 is the highest setting, wouldn't setting all three pins high bring it to 1/32 mode? You might want to double check that table for accuracy.

Thank you again!

Mike Sims

Reply

> Benne de Bakker says
> June 13, 2019 at 6:04 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-100)
>
> Hi Mike,
>
> Thanks for the feedback! I think I accidentally copied the shortcodes for the A4988 microstepping table when updating this post, so the 1/16 and 1/32 microstep settings were indeed not correct. The post should now be showing the correct table.
>
> Thanks again!
>
> Benne
>
> Reply

MAS says
July 12, 2019 at 4:36 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-245)

hello, VERY helpful example / tutorial. Thank you. I do have a question on acceleration usage. I am using 2 stepper motors and with your help it is working great. I also use "multistepper" to run them at the same time.
Now I want to add "acceleration" to each one. Is this possible, your comment about blocking confuses me. Do you have a suggestion on 2 steppers using acceleration and multi stepper. I am trying to make a pan / tilt fixture.

Reply

Gad (http://www.tekxul.com) says
July 27, 2019 at 8:44 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-306)

Hi Mike,
This information is very helpful and useful. I want to enable and disable the stepper motor, how should i go about it?
Thank you.
Regards,
Gad

Reply

Benne de Bakker says
August 15, 2019 at 7:42 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-391)

Hi Gad,

You can connect the enable pin of the driver to one of the digital outputs of the Arduino. If you set this pin HIGH, the stepper motor is disabled. By default, it is pulled low (also if you don't connect it to anything), which enables the driver/stepper motor.

Check the pinout at the beginning of the article for the location of the enable pin (EN).

Benne

Reply

Lorne C. says
August 1, 2019 at 4:55 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-325)

This tutorial was very helpful! Thank you so much.

I'm trying to make a rodent maze with automated doors. Everything works great except I'm unable to make the motor go faster than the 1000microsecond code (if that makes sense). As soon as I change the speed to below 1000microseconds, my motor won't move unless I force it. Any suggestions?

Reply

Benne de Bakker says
August 15, 2019 at 7:39 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-390)

Hi Lorne,

If your motor is stalling, I would try to increase the current limit slightly. You can also try to use a higher voltage power supply, which should also allow you to achieve higher speeds. What microstepping settings are you using?

Benne

Reply

HMuller says
August 14, 2019 at 7:54 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-387)

Hello,

I was playing with the "2. Example code to control number of steps or revolutions" and found that I could easily block the stepper. When blocked it still reached the end position in software, but not in reality. So you cannot trust the position this way, right? So if the position is important, I need an external sensor (eg quadrature encoder). Do you agree?

Or is there a way to control a stepper in such a way that you have position feedback that you can really trust?

Reply

Benne de Bakker says
August 15, 2019 at 7:24 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-389)

Hi,

If your stepper motor easily 'loses steps', I would increase the current limit slightly. The maximum speed you can reach (without losing steps or stalling the motor) also depends on the power supply and current limit that you are using.

You could look into a closed-loop stepper motor system (with an encoder) or even servomotors, but this would be much more expensive. In most cases, it is best to make sure that your motor can provide enough torque and speed so it doesn't lose steps and also incorporate a homing procedure in your setup. This allows you to reset/recalibrate the motor automatically.

Benne

Reply

Trackbacks

How to Control a Stepper Motor with Arduino Motor Shield (Rev 3) (https://www.makerguides.com/tutorials/arduino-motor-shield-stepper-motor/) says:
May 8, 2019 at 11:04 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-34)
[...] [...]

Rideaux Automatisés Ép.1: Test ESP8266 + Moteur pas à pas - TD72PRO (https://td72pro.com/rideaux-automatises-ep-1-test-esp8266-moteur-pas-a-pas/) says:
May 11, 2019 at 4:01 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-43)
[...] Contrôler un stepper motor avec Arduino: https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/ (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/) [...]

How to Control a Stepper Motor with A4988 and Arduino (4 Examples) (https://www.makerguides.com/a4988-stepper-motor-driver-arduino-tutorial/) says:
May 20, 2019 at 8:12 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-64)
[...] you have an DRV8825 stepper motor driver instead of the A4988 or want to know different ways to control stepper motors, [...]

Rideaux Automatisés Ép.5: Programmation du système et de l'interface utilisateur! - TD72PRO (https://td72pro.com/rideaux-automatises-ep-5-programmation-du-systeme-et-de-linterface-utilisateur/) says:
August 12, 2019 at 4:18 pm (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-372)
[...] https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/ (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/) [...]

Stepper Motor with L298N and Arduino Tutorial (4 Examples) (https://www.makerguides.com/l298n-stepper-motor-arduino-tutorial/) says:
August 18, 2019 at 9:43 am (https://www.makerguides.com/drv8825-stepper-motor-driver-arduino-tutorial/#comment-410)
[...] L298N motor driver, it is best to use a chopper drive instead. I wrote tutorials for the A4988 and DRV8825 driver that work great with many stepper [...]



(https://www.makerguides.com/dht11-dht22-arduino-tutorial/)

How to use DHT11 and DHT22 Sensors with Arduino (https://www.makerguides.com/dht11-dht22-arduino-tutorial/)



(https://www.makerguides.com/fsr-arduino-tutorial/)

Force Sensing Resistor (FSR) with Arduino Tutorial (https://www.makerguides.com/fsr-arduino-tutorial/)



(https://www.makerguides.com/hc-sr501-arduino-tutorial/)

How to use HC-SR501 PIR Motion Sensor with Arduino (https://www.makerguides.com/hc-sr501-arduino-tutorial/)